Меметичні алгоритми
Меметика (memetics) є підхід до моделей еволюційної передачі інформації, заснований на концепції мемів. Меми, що є аналогами генів, є одиницями культурної інформації, що поширюється між людьми через імітацію, навчання та інші засоби. Поняття мему було введено та основи меметики були розроблені Докінзом (C.R. Dawkins) у 1976 році. Меми можуть передаватися вертикально – від попередників, батьків, вихователів, книг, культурних артефактів тощо. Також можлива горизонтальна передача мемів від людини до людини та культури до культури. Незважаючи на те, що меми є чистою інформацією, їх функціонування призводить до істотних змін поведінки людини.
Меметичні алгоритми (Memetic algorithms, М-алгоритми) визначаються як гібридні популяційні метаевристичні алгоритми пошукової оптимізації, засновані на концепції мему та неодарвінівському принципі еволюції. У контексті М-алгоритмів мем є реалізацією алгоритму локальної оптимізації, який уточнює поточні рішення на кожній ітерації або через кілька ітерацій. М-алгоритми можна розглядати як гібридизацію одного з популяційних алгоритмів глобального пошуку та одного або кількох класичних або популяційних алгоритмів локальної оптимізації. Спочатку М-алгоритми були запропоновані як один із варіантів підвищення ефективності генетичних алгоритмів.

Ефективність М-алгоритмів значною мірою залежить від значень їх вільних властивостей. У низці досліджень показано, що дуже великий вплив на ефективність цих алгоритмів надає вибір використовуваних мемів. Тому ця проблема займає одне з центральних місць у роботах, присвячених М-алгоритмам.

Меми є однією з революційних ідей, що передбачає схожість еволюції генів і еволюції людської культури. Докінз назвав одиницю культурного обміну, аналог гена у культурі, мемом. Мемом можна назвати жест, слово чи ідею; Будь-яка одиниця культурної інформації, яка може бути передана від людини до людини за допомогою імітації навчання – це мем.
Меметичні алгоритми, на відміну генетичних, імітують процес культурної еволюції. Розроблений Москато підхід використовує аналогію з еволюцією мемов. Меметичний алгоритм складається з наступних етапів: локальний пошук, кооперація, змагання та критерій закінчення пошуку. Меметичним називають широкий клас алгоритмів, об'єднаних загальною ідеєю: включенням до генетичного алгоритму індивідуального навчання особин та використання інформації про структуру простору можливих рішень. Проблема балансу між популяційним та локальним пошуком можна розглядати як загальну проблему дотримання балансу між екстенсивними та інтенсивними дослідженнями простору пошуку.

Меметичний алгоритм у загальних рисах включає такі компоненти:

1. Локальний пошук. Для його здійснення можна використовувати алгоритм імітації відпалу та алгоритми підйому.

2. Кооперація. Обмін інформацією між особами може здійснюватися через процедуру, аналогічну до застосування оператора двоточкового схрещування в генетичних алгоритмах.

3. Змагання. Процедура селекції аналогічна до генетичних алгоритмів і зазвичай полягає у відборі найбільш пристосованих особин у популяції та виключенні з неї менш пристосованих.

4. Критерій закінчення пошуку. Меметичні алгоритми можуть включати оцінку різноманітності особин, крім підрахунку числа ітерацій і оцінки поліпшення результату.
Меметичні алгоритми – це широкий клас алгоритмів, об'єднаних загальною ідеєю індивідуального навчання особин та використання інформації про структуру простору можливих рішень.

Меми, подібно до генів, є реплікаторами, тобто об'єктами, здатними до самовідтворення. Для мемів виживання та відтворення залежать від наявності носія, який поширює мем. Меми можуть видозмінюватися, формуючи нові меми, і беруть участь у боротьбі ресурси - уми людей.

Меми часто об'єднуються в комплекси, або мемплекси, для посилення у боротьбі за носіїв. Мемплекси аналогічні симбіотичним сукупностям генів, що становлять генетичні коди біологічних організмів. Прикладом мемплексу може бути релігія. Мемплекси надають глибокий вплив на формування індивідуальної та суспільної поведінки.
Алгоритм плигаючих жаб

Тасуючий алгоритм стрибаючих жаб (Shuffled Frog Leaping Algorithm, SFL) був запропонований Юсуфом (М. Eusuff) та іншими авторами у 2003 році. Цей алгоритм поєднує принципи меметичного алгоритму та алгоритму рою часток, його розробка була натхненна поведінкою групи жаб у процесі пошуку їжі.

Спочатку алгоритм SFL розробили як метаевристичний метод на вирішення завдань комбінаторної оптимізації. Він ґрунтується на використанні математичних функцій та поінформованого евристичного пошуку.

Алгоритм SFL складається з кількох взаємодіючих віртуальних популяцій жаб, які називаються мемплексами. Віртуальні жаби виконують роль господарів чи носіїв мемів, де мем є одиницю культурної еволюції. У кожному мемплексі відбувається незалежний локальний пошук з використанням методу, аналогічного до оптимізації рою часток, але з упором на локальний пошук.

Для забезпечення глобальних досліджень віртуальні жаби періодично перетасовуються та реорганізуються у нові мемплекси за допомогою методу, аналогічного алгоритму перетасованої складної еволюції. Крім того, у популяції генеруються та замінюються випадкові віртуальні жаби, щоб забезпечити можливість випадкової генерації покращеної інформації.

Переходимо безпосередньо до тасуючого алгоритму жаб, що стрибають. Основна ідея алгоритму полягає у розбитті всієї популяції жаб з глобальним лідером G на мем - групи, в кожній з яких є лідер L (рисунок 1).

Рисунок 1. Населення жаб із глобальним лідером (G), розбита на меми зі своїми локальним лідером (L).
У межах кожного мему жаби прагнуть рухатися у напрямі локального лідера. Лідерство оновлюється, якщо позиція однієї із жаб покращується. Меми не поділяються у просторі пошуку і можуть перетинатись. Таким чином, жаба, що знаходиться на території одного мему, цілком може належати до іншого. Це створює динамічне середовище, де жаби можуть переходити просторово з одного мему до іншого, залежно від зміни їх позицій у просторі пошуку.

Розглядаємо завдання глобальної умовної оптимізації, у якій фітнес-функція підлягає максимізації. Алгоритм SFL включає наступні основні кроки.

1.0 Ініціалізація алгоритму

1.1 Створення початкової популяції жаб із випадковими координатами.

1.2 Визначення пристосованості кожної жаби.

1.3 Створення мемплексів (підпопуляцій), розподіл жаб за мемплексами.
2.0 Цикл пошуку оптимального рішення

2.1 Для кожного мемплексу:

 2.1.1 Визначити найкращу жабу.

 2.1.2 Перемістити інші жаби у бік кращої жаби в мемплексі.

2.2 Якщо переміщення жаби не покращило її пристосованість:

 2.2.1 Перемістити її у бік глобально кращої жаби.

 2.2.2 Якщо це не покращило пристосованість, перемістити жабу у випадкове місце на полі.

3.0 Вимірювання пристосованості жаб

3.1 Для кожної жаби у популяції виміряти її пристосованість.

3.2. Оновити інформацію про кращу жабу в кожному мемплексі та популяції в цілому.

4.0 Визначення глобально кращої жаби та оновлення мемплексів

4.1 Визначити глобально найкращу жабу по всій популяції.

4.2 Якщо це останній цикл:

 4.2.1 Скинути лічильник циклів мемплексу.

 4.2.2. Згенерувати випадкові індекси для жаб.

 4.2.3 Скопіювати жаб у мемплекси за новими індексами.

 4.2.4 Визначити у кожному мемплексі найкращу жабу.

 4.2.5 Скинути пристосованість жаб та крок.

4.3 Якщо це останній цикл:

 4.3.1 Скопіювати пристосованість та координати жаб із популяції у відповідні жаби мемплексів.

 4.3.2 Визначити у кожному мемплексі найкращу жабу.

 4.3.3 Визначити напрямок наступного стрибка для кожної жаби залежно від її пристосованості.
Таким чином, основою алгоритму SFL є комбінування локального пошуку в межах кожного з мемеплексів та глобального пошуку шляхом обміну інформацією про положення кращих агентів мемеплексів.

Даний алгоритм є модифікацією алгоритму SFL, в якому на етапі локального пошуку агент рухається не точно в напрямку кращого агента відповідного мемеплексу, а в деякому випадково обраному напрямку. Відомі як послідовні, і паралельні гібридизації алгоритму з багатьма популяційними алгоритмами, наприклад з алгоритмом штучної імунної системи.

Рисунок 2. Стрибки жаб. Якщо попереднє переміщення виявилося невдалим, то наступний стрибок буде зроблено з того самого місця.
Логічною одиницею алгоритму оптимізації SFL є жаба, вона може бути описана структурою S_Frog, яка є агентом у просторі пошуку.

Структура S_Frog містить такі поля:

c – масив координат поточної позиції жаби.

cPrev – масив координат попередньої позиції жаби.

f – значення функції пристосованості для поточної позиції жаби.

fPrev – значення функції пристосованості для попередньої позиції жаби.

frogStep – номер кроку, на якому знаходиться жаба.
//——————————————————————————————————————————————————————————————————————————————
struct S_Frog

{

  void Init (int coords)

  {

    ArrayResize (c,     coords);

    ArrayResize (cPrev, coords);

    f        = -DBL_MAX;

    fPrev    = -DBL_MAX;

    frogStep = 0;

  }

  double c     []; //coordinates
  double cPrev []; //previous coordinates
  double f;        //fitness
  double fPrev;    //previous fitness
  int    frogStep; //frog step
};

//——————————————————————————————————————————————————————————————————————————————
Структура S_Memeplex визначає мемплекс в алгоритмі. Вона містить такі поля:

frogs – масив жаб, що становлять мемплекс.

fBest – найкраще значення функції пристосованості серед усіх жаб у мемплексі.

cBest – масив координат, що відповідають кращому значенню функції пристосованості у мемплексі.
//——————————————————————————————————————————————————————————————————————————————
struct S_Memeplex

{

  S_Frog frogs [];

  double fBest;     //best fitness
  double cBest [];  //best coordinates
};

//——————————————————————————————————————————————————————————————————————————————
Клас C_AO_SFL надає методи ініціалізації алгоритму, переміщення жаб і ревізії популяції. Він також містить допоміжні методи для перетворення значень та генерації випадкових чисел.

Напишемо клас алгоритму SFL, що включає поля:

cB – масив, що зберігає найкращі координати (best coordinates);

fB – змінна, що зберігає значення цільової функції (fitness function) кращих координат;

frogs – масив, що зберігає всіх жаб у популяції;

mems – масив, що зберігає мемплекси (групи жаб);

rangeMax – масив, що зберігає максимальні значення кожної координати пошуку;

rangeMin – масив, що зберігає мінімальні значення кожної координати пошуку;

rangeStep – масив, що зберігає кроки пошуку кожної координати.
Відкриті методи класу:

Init – метод ініціалізації параметрів алгоритму, що приймає наступні параметри:

coordinatesNumberP – кількість координат пошуку;

populationSizeP – розмір популяції (кількість жаб);

numbMemsP – кількість мемплексів (груп жаб);

numbCyclesP – кількість циклів у мемплексі;

frogStepsToLocalMaxP – кількість кроків жаби до локального максимуму;

movConstantP – константа переміщення (крок пошуку) жаб.

Moving – метод, що реалізує переміщення жаб у просторі пошуку.

Revision – метод, що реалізує ревізію популяції жаб та оновлення найкращих координат.

SeInDiSp – допоміжний метод, що реалізує перетворення значення з одного діапазону в інший із заданим кроком.

RNDfromCI – допоміжний метод, що генерує випадкове число у заданому інтервалі.
Опис приватних полів класу:

coordinatesNumber – кількість координат пошуку;

frogsNumber – кількість жаб у популяції;

numbMems – кількість мемплексів (груп жаб);

numbCycles – кількість циклів у мемплексі;

frogStepsToLocalMax – кількість кроків жаби до локального максимуму;

movConstant – константа переміщення (крок пошуку) жаб;

memsFrogs – кількість жаб у кожному мемплексі;

numbCyclesCNT – лічильник циклів;

vect – масив, що зберігає вектор;

indexes – масив, що зберігає індекси;

revision – прапор, що вказує на необхідність ревізії популяції.
//——————————————————————————————————————————————————————————————————————————————
class C_AO_SFL

{

  //----------------------------------------------------------------------------
  public: double cB        []; //best coordinates
  public: double fB;           //FF of the best coordinates
  public: S_Frog frogs     []; //all frogs
  public: S_Memeplex mems  []; //memeplexes
  public: double rangeMax  []; //maximum search range
  public: double rangeMin  []; //manimum search range
  public: double rangeStep []; //step search
  public: void Init (const int    coordinatesNumberP,   //coordinates number
                     const int    populationSizeP,      //population size
                     const int    numbMemsP,            //number of memeplexes
                     const int    numbCyclesP,          //number of cycles in the memeplex
                     const int    frogStepsToLocalMaxP, //frog steps to the local maximum
                     const double movConstantP);        //movement step (0.0 .. 1.0)
  public: void Moving   ();

  public: void Revision ();

  //----------------------------------------------------------------------------
  private: int    coordinatesNumber;   //coordinates number
  private: int    frogsNumber;         //frogs number
  private: int    numbMems;            //number of memeplexes
  private: int    numbCycles;          //number of cycles in the memeplex
  private: int    frogStepsToLocalMax; //frog steps to the local maximum
  private: double movConstant;         //movement step (0.0 .. 1.0)
  private: int    memsFrogs;           //number of frogs in the memplex
  private: int    numbCyclesCNT;       //cycle counter
  private: double vect    [];          //vector
  private: int    indexes [];          //indexes
  private: bool   revision;

  private: double SeInDiSp  (double In, double InMin, double InMax, double Step);

  private: double RNDfromCI (double min, double max);

};

//——————————————————————————————————————————————————————————————————————————————
Для ініціалізації алгоритму будемо використовувати метод ініціалізації Init, який має кілька параметрів:

coordinatesNumberP – кількість координат

populationSizeP – розмір популяції

numbMemsP – кількість мемплексів

numbCyclesP – кількість циклів у мемплексі

frogStepsToLocalMaxP – кількість кроків жаби до локального максимуму

movConstantP – крок переміщення (від 0.0 до 1.0)

Спочатку метод скидає генератор випадкових чисел і встановлює початкове значення змінної fB (-DBL_MAX) та revision (false).

Потім метод створює масиви rangeMax, rangeMin, rangeStep, vect, indexes, cB, frogs та mems з потрібними розмірами.

Метод ініціалізує кожну жабу в масиві frogs та кожну жабу в кожному мемплексі в масиві mems за допомогою методу Init, передаючи кількість координат.

Потім метод встановлює початкове значення змінної fBest кожного мемплексу -DBL_MAX і створює масив cBest для кожного мемплексу з потрібним розміром.
//——————————————————————————————————————————————————————————————————————————————
void C_AO_SFL::Init (const int    coordinatesNumberP,   //coordinates number
                     const int    populationSizeP,      //population size
                     const int    numbMemsP,            //number of memeplexes
                     const int    numbCyclesP,          //number of cycles in the memeplex
                     const int    frogStepsToLocalMaxP, //frog steps to the local maximum
                     const double movConstantP)         //movement step (0.0 .. 1.0)
{

  MathSrand ((int)GetMicrosecondCount ()); // reset of the generator
  fB       = -DBL_MAX;

  revision = false;

  coordinatesNumber   = coordinatesNumberP;

  frogsNumber         = populationSizeP;

  numbMems            = numbMemsP;

  numbCycles          = numbCyclesP;

  frogStepsToLocalMax = frogStepsToLocalMaxP;

  movConstant         = movConstantP;

  memsFrogs           = frogsNumber / numbMems;

  numbCyclesCNT       = 0;

  ArrayResize (rangeMax,  coordinatesNumber);

  ArrayResize (rangeMin,  coordinatesNumber);

  ArrayResize (rangeStep, coordinatesNumber);

  ArrayResize (vect,      coordinatesNumber);

  ArrayResize (indexes,   frogsNumber);

  ArrayResize (cB, coordinatesNumber);

  ArrayResize (frogs, frogsNumber);

  for (int i = 0; i < frogsNumber; i++)

  {

    frogs [i].Init (coordinatesNumber);

  }

  ArrayResize (mems, numbMems);

  for (int i = 0; i < numbMems; i++)

  {

    ArrayResize (mems [i].frogs, memsFrogs);

    for (int frgs = 0; frgs < memsFrogs; frgs++)

    {

      mems [i].frogs [frgs].Init (coordinatesNumber);

    }

    mems [i].fBest = -DBL_MAX;

    ArrayResize (mems [i].cBest, coordinatesNumber);

  }

}

//——————————————————————————————————————————————————————————————————————————————
Метод Moving призначений для переміщення жаб простором пошуку. Метод досить великий та розберемо його частинами.

На початку коду перевіряється значення змінної revision. Якщо воно дорівнює false, то виконується наступний блок коду:

- Задається початкове значення змінної fB, яка є найкращою оцінкою функції. У разі їй присвоюється значення -DBL_MAX (негативна нескінченність).

- Запускається цикл, у якому відбувається ініціалізація кожної жаби. Для кожної жаби:

- Запускається цикл з кожної координати c.

- Генерується випадкове значення координати c за допомогою функції RNDfromCI, яка повертає випадкове число в заданому діапазоні.

- Значення координати c наводиться до діапазону за допомогою функції SeInDiSp, яка дозволяє зрушити значення всередині діапазону з певним кроком.

- Значення функції f для жаби встановлюється -DBL_MAX.

- Значення попереднього значення fPrev для жаби встановлюється рівним -DBL_MAX.

- Крок жаби frogStep встановлюється рівним 0.

- Запускається цикл з кожної координати c.

- Обчислюється значення vect [c], яке є добутком різниці між максимальним і мінімальним значенням діапазону на movConstant.

- Змінної revision надається значення true, щоб вказати, що ініціалізація була виконана.

- Змінної numbCyclesCNT надається значення 0.
Таким чином, код виконує ініціалізацію жаб, встановлює початкові значення функції та інших параметрів, а також обчислює значення vect [c] для кожної координати c.
if (!revision)

{

  fB = -DBL_MAX;

  for (int frgs = 0; frgs < frogsNumber; frgs++)

  {

    for (int c = 0; c < coordinatesNumber; c++)

    {

      frogs [frgs].c [c] = RNDfromCI (rangeMin [c], rangeMax [c]);

      frogs [frgs].c [c] = SeInDiSp (frogs [frgs].c [c], rangeMin [c], rangeMax [c], rangeStep [c]);

      frogs [frgs].f        = -DBL_MAX;

      frogs [frgs].fPrev    = -DBL_MAX;

      frogs [frgs].frogStep = 0;

    }

  }

  for (int c = 0; c < coordinatesNumber; c++)

  {

    vect [c] = (rangeMax [c] - rangeMin [c]) * movConstant;

  }

  revision      = true;

  numbCyclesCNT = 0;

}

Якщо прапор revision дорівнює true, це означає, що алгоритм перейшов у стадію переміщення жаб. Зі змінними методу ми вже познайомилися, тому докладно зупинятися на них не будемо. У цій частині коду реалізуються стрибки жаб відповідно до індивідуального для кожної жаби кроку. На першому кроці жаба стрибає у бік локального лідера, спроба зараховується, якщо положення жаби стало краще, інакше лічильник кроку буде збільшено. Тобто на стрибки у бік локального лідера виділяється фіксована кількість спроб відповідно до зовнішніх параметрів алгоритму.

Для жаби-лідера використовується інший принцип переміщення на відміну від інших у мемплексі. Лідер просто робить стрибки у випадковому напрямку у невеликому околі свого положення.

На відміну від лідера всі інші жаби роблять стрибки у бік лідера за такою формулою:
coord = mems [m]. frogs [frgs]. cPrev [c] + rnd * vect [c] * ((mems [m]. cBest [c] - mems [m]. frogs [frgs]. ) / eDistance);

З формули видно, що нова координата жаби буде отримана шляхом переміщення у бік лідера на відстань між ними, нормована на евклідову відстань по всіх координатах, при цьому вводиться випадкова компонента rnd.

else
{

  int cnt = 0;

  double eDistance = 0.0; //euclidean distance
  double coordDiff = 0.0; //the difference in coordinates
  for  (int m = 0; m < numbMems; m++)

  {

    for (int frgs = 0; frgs < memsFrogs; frgs++)

    {

      //2.1 передвинуть лягушки в сторону лучшего в мемплексе-----------------
      if (mems [m].frogs [frgs].frogStep < frogStepsToLocalMax)

      {

        if (mems [m].frogs [frgs].fPrev != -DBL_MAX && mems [m].fBest != -DBL_MAX)

        {

          if (mems [m].frogs [frgs].fPrev == mems [m].fBest)

          {

            for (int c = 0; c < coordinatesNumber; c++)

            {

              rnd = RNDfromCI (-1.0, 1.0);

              rnd = rnd < 0.0 ? -rnd * rnd : rnd * rnd;

              coord = mems [m].frogs [frgs].cPrev [c] + rnd * vect [c] * 0.2;

              mems [m].frogs [frgs].c [c] = SeInDiSp (coord, rangeMin [c], rangeMax [c], rangeStep [c]);

            }

          }

          else
          {

            eDistance = 0.0;

            coordDiff = 0.0;

            //посчитаем евклидово расстояние----------------------------------
            for (int c = 0; c < coordinatesNumber; c++)

            {

              coordDiff  = mems [m].cBest [c] - mems [m].frogs [frgs].cPrev [c];

              coordDiff *= coordDiff;

              eDistance += coordDiff;

            }

            eDistance = sqrt (eDistance);

            for (int c = 0; c < coordinatesNumber; c++)

            {

              rnd = RNDfromCI (-1.0, 1.0);

              coord = mems [m].frogs [frgs].cPrev [c] + rnd * vect [c] * ((mems [m].cBest [c] - mems [m].frogs [frgs].cPrev [c]) / eDistance);

              mems [m].frogs [frgs].c [c] = SeInDiSp (coord, rangeMin [c], rangeMax [c], rangeStep [c]);

            }

          }

        }

        else
        {

          for (int c = 0; c < coordinatesNumber; c++)

          {

            rnd = RNDfromCI (-1.0, 1.0);

            rnd = rnd < 0.0 ? -rnd * rnd : rnd * rnd;

            coord = mems [m].frogs [frgs].cPrev [c] + rnd * vect [c];

            mems [m].frogs [frgs].c [c] = SeInDiSp (coord, rangeMin [c], rangeMax [c], rangeStep [c]);

          }

        }

      }

      else
      {

        //2.2 передвинуть в сторону глобального если ход хуже предыдущего-----
        if (mems [m].frogs [frgs].frogStep /*==*/ >= frogStepsToLocalMax)

        {

          eDistance = 0.0;

          coordDiff = 0.0;

          //посчитаем евклидово расстояние------------------------------------
          for (int c = 0; c < coordinatesNumber; c++)

          {

            coordDiff  = cB [c] - mems [m].frogs [frgs].cPrev [c];

            coordDiff *= coordDiff;

            eDistance += coordDiff;

          }

          eDistance = sqrt (eDistance);

          for (int c = 0; c < coordinatesNumber; c++)

          {

            rnd = RNDfromCI (-1.0, 1.0);

            coord = mems [m].frogs [frgs].cPrev [c] + rnd * vect [c] * ((cB [c] - mems [m].frogs [frgs].cPrev [c]) / eDistance);

            mems [m].frogs [frgs].c [c] = SeInDiSp (coord, rangeMin [c], rangeMax [c], rangeStep [c]);

          }

        }

        //2.3 передвинуть случайно по полю --------------------------------
        else
        {

          for (int c = 0; c < coordinatesNumber; c++)

          {

            rnd = RNDfromCI (-1.0, 1.0);

            coord    = RNDfromCI (rangeMin [c], rangeMax [c]);

            mems [m].frogs [frgs].c [c] = SeInDiSp (coord, rangeMin [c], rangeMax [c], rangeStep [c]);

          }

        }

      }

      frogs [cnt] = mems [m].frogs [frgs];

      cnt++;

    }

  }

  //--------------------------------------------------------------------------
  numbCyclesCNT++;

}

Метод Revision призначений для визначення глобального лідера кожної ітерації алгоритму і визначення локальних лідерів. Якщо кількість циклів, відведених на переміщення жаб у межах кожного мемплексу, вичерпано, то проводимо тасування – перепризначаємо жаби по мемплексам випадковим чином, цим здійснюється обмін інформацією між мемплексами. У цьому методі так само проводиться облік відповідних кроків стрибків – куди переміщатиметься жаба на наступній ітерації (у бік локального чи глобального лідера або у випадковому напрямку у просторі пошуку).

//——————————————————————————————————————————————————————————————————————————————
void C_AO_SFL::Revision ()

{

  //4.1 определить глобально лучшего по популяции-------------------------------
  for (int i = 0; i < frogsNumber; i++)

  {

    if (frogs [i].f > fB)

    {

      fB = frogs [i].f;

      ArrayCopy (cB, frogs [i].c, 0, 0, WHOLE_ARRAY);

    }

  }

  int cnt = 0;

  //если последний цикл=========================================================
  if (numbCyclesCNT >= numbCycles)

  {

    //4.2.0 сбросить счетчик циклов мемплекса-----------------------------------
    numbCyclesCNT = 0;

    //4.2.1 сгенерировать случайно индексы--------------------------------------
    for (int i = 0; i < frogsNumber; i++) indexes [i] = i;

    Shuffle (indexes, frogsNumber);

    //4.2.2 скопировать случайно в мемплексы------------------------------------
    for  (int m = 0; m < numbMems; m++)

    {

      mems [m].fBest = -DBL_MAX;

      for (int frgs = 0; frgs < memsFrogs; frgs++)

      {

        mems [m].frogs [frgs] = frogs [indexes [cnt]];

        cnt++;

        //4.2.3 определить в каждом мемплексе лучшего---------------------------
        if (mems [m].frogs [frgs].f > mems [m].fBest)

        {

          mems [m].fBest = mems [m].frogs [frgs].f;

          ArrayCopy (mems [m].cBest, mems [m].frogs [frgs].c, 0, 0, WHOLE_ARRAY);

        }

        //4.2.4 сбросить приспособленность лягушек и шаг------------------------
        mems [m].frogs [frgs].f        = -DBL_MAX;

        mems [m].frogs [frgs].fPrev    = -DBL_MAX;

        mems [m].frogs [frgs].frogStep = 0;

      }

    }

  }

  //если НЕ последний цикл======================================================
  else
  {

    for  (int m = 0; m < numbMems; m++)

    {

      for (int frgs = 0; frgs < memsFrogs; frgs++)

      {

        mems [m].frogs [frgs].frogStep++;

        //4.3.1 скопировать приспособленность и координаты лягушек из популяции в
        //соответствующие лягушки мемплексов
        mems [m].frogs [frgs].f = frogs [cnt].f;

        ArrayCopy (mems [m].frogs [frgs].c, frogs [cnt].c, 0, 0, WHOLE_ARRAY);

        //4.3.2 определить в каждом мемплексе лучшего---------------------------
        if (frogs [cnt].f > mems [m].fBest)

        {

          mems [m].fBest = frogs [cnt].f;

          ArrayCopy (mems [m].cBest, frogs [cnt].c, 0, 0, WHOLE_ARRAY);

        }

        //4.3.3 определить направление следующего прыжка------------------------
        if (mems [m].frogs [frgs].frogStep <= frogStepsToLocalMax)

        {

          if (mems [m].frogs [frgs].f > mems [m].frogs [frgs].fPrev)

          {

            mems [m].frogs [frgs].fPrev = mems [m].frogs [frgs].f;

            ArrayCopy (mems [m].frogs [frgs].cPrev, mems [m].frogs [frgs].c, 0, 0, WHOLE_ARRAY);

            mems [m].frogs [frgs].frogStep = 0;

          }

        }

        else
        {

          if (mems [m].frogs [frgs].frogStep >= frogStepsToLocalMax + 1)

          {

            if (mems [m].frogs [frgs].f > mems [m].frogs [frgs].fPrev)

            {

              mems [m].frogs [frgs].fPrev = mems [m].frogs [frgs].f;

              ArrayCopy (mems [m].frogs [frgs].cPrev, mems [m].frogs [frgs].c, 0, 0, WHOLE_ARRAY);

              mems [m].frogs [frgs].frogStep = 0;

            }

          }

          else
          {

            mems [m].frogs [frgs].f        = -DBL_MAX;

            mems [m].frogs [frgs].fPrev    = -DBL_MAX;

            mems [m].frogs [frgs].frogStep = 0;

          }

        }

        cnt++;

      }

    }

  }

}

//——————————————————————————————————————————————————————————————————————————————
В алгоритмі оптимізації SFL нам доведеться перемішувати жаби між мемплексами випадковим чином. Завдання випадкового тасування цікаве тим, що його алгоритмічне рішення нетривіальне, проте завдяки Рональду Фішеру та Френку Йєтсу світ отримав ефективний і швидкий алгоритм. Він широко використовується в комп'ютерних науках, особливо в галузі генетичних алгоритмів, криптографії та статистики.

Основні переваги алгоритму Фішера - Йєтса:

1. Простота реалізації: Алгоритм Фішера - Йєтса легко реалізувати будь-якою мовою програмування. Він вимагає складних математичних обчислень чи спеціальних бібліотек.

2. Ефективність: Алгоритм Фішера - Йєтса працює за лінійний час, тобто час виконання залежить від кількості елементів, які потрібно переставити. Це робить його дуже ефективним для великих наборів даних.

3. Випадковість: Алгоритм Фішера - Йєтса забезпечує високу випадковість перестановок. Це важливо для багатьох програм, таких як випадкові тести, шифрування та симуляції.

4. Незалежність від вхідних даних: Алгоритм Фішера-Йетса може бути застосований до будь-якого набору даних без необхідності знання їх структури або властивостей. Це робить його універсальним інструментом для багатьох завдань.

5. Псевдовипадковість: Алгоритм Фішера - Йєтса генерує псевдовипадкові перестановки, які можуть бути використані в багатьох додатках, де потрібна випадковість, але не обов'язково справжня випадковість.

Загалом алгоритм Фішера — Йєтса цікавий своєю простотою, ефективністю та універсальністю. Він є корисним інструментом для багатьох завдань, пов'язаних із випадковістю та перестановками даних.
/——————————————————————————————————————————————————————————————————————————————
void Shuffle (int & arr [], int size)

{

  int index, temp;

  for (int i = size - 1; i > 0; i--)

  {

    index = MathRand () % (i + 1);

    temp = arr [i];

    arr [i] = arr [index];

    arr [index] = temp;

  }

}

//——————————————————————————————————————————————————————————————————————————————
Тестування

Роздрукування роботи тасуючого алгоритму жаб, що стрибають, на тестовому стенді:
C_AO_SFL:50;25;15;5;0.7
=============================
5 Rastrigin's; Func runs 10000 result: 66.52578871476199
Score: 0.82429
25 Rastrigin's; Func runs 10000 result: 52.38937199890908
Score: 0.64913
500 Rastrigin's; Func runs 10000 result: 44.5591163558836
Score: 0.55211
=============================
5 Forest's; Func runs 10000 result: 0.5762718670482314
Score: 0.32597
25 Forest's; Func runs 10000 result: 0.16329693292687839
Score: 0.09237
500 Forest's; Func runs 10000 result: 0.04968320483668511
Score: 0.02810
=============================
5 Megacity's; Func runs 10000 result: 3.1599999999999997
Score: 0.26333
25 Megacity's; Func runs 10000 result: 1.016
Score: 0.08467
500 Megacity's; Func runs 10000 result: 0.3004
Score: 0.02503
=============================
All score: 2.84501
При спостереженні за анімацією роботи алгоритму SFL не виявляється жодної кластеризації чи групування за окремими мемами, хоча очікувалося протилежне. Агенти (точки) оптимізації розподілені по полю хаотично, і немає жодних закономірностей у русі частинок. Відразу помітно низька якість збіжності алгоритму, що, на жаль, проявляється у застряванні в локальних екстремумах, що можна визначити за довгими плавними ділянками на графіку збіжності. Однак зі збільшенням кількості оптимізованих параметрів "ступінчастість" знижується.

